MAT4161/MAT4181 - Cálculo a uma Variável - 2024.2

P1 Maple - Versão II - Gabarito

Questão 1

Seja $f : \mathbb{R} \to \mathbb{R}$ a função definida por

$$f(x) = -2x^4 - 10x^3 - 10x^2 .$$

(a) Determine a equação de uma reta r, que é tangente ao gráfico de f e possui inclinação 5.

Solução:

```
> restart
> f:=x->-2*x^4-10*x^3-10*x^2
```

$$f := x \mapsto -2 \cdot x^4 - 10 \cdot x^3 - 10 \cdot x^2 \tag{1}$$

```
> a:=fsolve(D(f)(x)=5)
```

$$a := -2.981897941 \tag{2}$$

```
> D(f)(a)*(x-a)+f(a)
```

$$5.00000002\, x + 33.00900985 \tag{3}$$

A equação da reta r é então $y = -5\,x + 33.009$

(b) O gráfico de f possui alguma outra reta tangente de inclinação 5? Justifique sua resposta.

Solução:

```
> solve(D(f)(x)=5)
```

$$-\frac{\left(1215 + 30\sqrt{354}\right)^{1/3}}{12} - \frac{35}{4\left(1215 + 30\sqrt{354}\right)^{1/3}} - \frac{5}{4}, \frac{\left(1215 + 30\sqrt{354}\right)^{1/3}}{24} + \frac{35}{8\left(1215 + 30\sqrt{354}\right)^{1/3}} - \frac{5}{4} + \frac{I\sqrt{3}\left(-\frac{\left(1215 + 30\sqrt{354}\right)^{1/3}}{12} + \frac{35}{4\left(1215 + 30\sqrt{354}\right)^{1/3}}\right)}{2}, \tag{4}$$

$$\frac{\left(1215+30\sqrt{354}\right)^{1/3}}{24}+\frac{35}{8\left(1215+30\sqrt{354}\right)^{1/3}}-\frac{5}{4}$$
$$-\frac{I\sqrt{3}\left(-\frac{\left(1215+30\sqrt{354}\right)^{1/3}}{12}+\frac{35}{4\left(1215+30\sqrt{354}\right)^{1/3}}\right)}{2}$$

A resposta então é não, pois as outras soluções para a equação $f\,'(x)=5$ são complexas.

(c) Seja s a reta tangente ao gráfico de f em $x=\frac{1}{3}$.

Determine a equação de uma reta t, que é paralela a s e também é tangente ao gráfico de f.

```
> sol:=solve(D(f)(x)=D(f)(1/3))
```

$$sol := \frac{1}{3}, -\frac{49}{24}-\frac{\sqrt{177}}{24}, -\frac{49}{24}+\frac{\sqrt{177}}{24} \qquad \textbf{(5)}$$

Pelo solve acima, temos que o gráfico de f possui outras duas retas tangentes que são paralelas a s. Como a questão pede apenas uma, vamos escolher qualquer uma delas:

```
> b:=evalf(sol[2])
```

$$b := -2.596005613 \qquad \textbf{(6)}$$

```
> D(f)(b)*(x-b)+f(b)
```

$$-10.29629624\,x-10.00536224 \qquad \textbf{(7)}$$

Logo, a equação de uma reta tangente ao gráfico de f e paralela a s é $y=-10.296\,x-10.005$.

Obs.: Caso tivéssemos escolhido a outra solução, teríamos o seguinte:

```
> c:=evalf(sol[3])
```

$$c := -1.487327721 \qquad \textbf{(8)}$$

```
> D(f)(c)*(x-c)+f(c)
```

$$-10.29629630\,x-14.32073713 \qquad \textbf{(9)}$$

Logo, a equação da outra reta tangente ao gráfico de f e paralela a s é $y=-10.296\,x-14.320$.

Questão 2

Considere as funções

$$f(x) = 2x^2 - x + 5 \qquad \text{e} \qquad g(x) = -x^3 - \frac{x}{2} - 3\,,$$

ambas com domínio $[-2, 2]$.

Seja $L(x)$ a distância vertical entre os dois gráficos (ou seja, para cada $x_0 \in [-2, 2]$, temos que $L(x_0)$ é a distância entre os pontos $(x_0, f(x_0))$ e $(x_0, g(x_0))$).

(a) Determine os valores de x para os quais a função L possui máximo local.

Solução:

```
> restart
> f:=x->2*x^2-x+5
```

$$f := x \mapsto 2 \cdot x^2 - x + 5 \qquad \textbf{(10)}$$

```
> g:=x->-x^3-x/2-3
```

$$g := x \mapsto -x^3 - \frac{1}{2} \cdot x - 3 \qquad \textbf{(11)}$$

```
> plot([f(x),g(x)],x=-2..2,color=[red,blue]);
```



Pelo plot acima vemos que, no intervalo $[-2, 2]$, o gráfico da função f está sempre acima do gráfico da função g. Logo, $L(x) = f(x) - g(x)$.

```
> L:=x->f(x)-g(x)
```

$$L := x \mapsto f(x) - g(x) \qquad \textbf{(12)}$$

```
> plot(L(x),x=-2..2)
```




```
> fsolve(D(L)(x)=0)
```

$$-1.448402627, 0.1150692933 \tag{13}$$

Pelo plot e pelo solve acima vemos que a função L possui máximo local em $x = -1.448$ e 2.

(b) Determine os valores de x para os quais a função L possui mínimo local.

Solução:

Utilizando o plot e o solve do item anterior, temos que a função L possui mínimo local em $x = -2$ e 0.115.

(c) Determine as coordenadas do ponto do gráfico de f e do ponto do gráfico de g para os quais a distância vertical entre os gráficos de f e g é **mínima**.

Solução:

Pelo plot acima, vemos que a função L possui mínimo global em $x = 0.115$.

```
> f(0.1150692933)
```

$$4.911412591 \tag{14}$$

```
> g(0.1150692933)
```

$$-3.059058273 \tag{15}$$

Logo, os pontos pedidos são:

No gráfico de f : $(0.115, 4.911\ )$

No gráfico de g : $(0.115,\ -3.059)$

Questão 3

Considere região plana $\mathcal{R}$ definida por

$$\mathcal{R} = \left\{(x, y) \in \mathbb{R}^2 \,\middle|\, 0 \le y \le \frac{x^3 + x^2 - 6x}{x^2 + 1} + 5\,,\ -1 \le x \le 1\right\}.$$

Seja $\mathcal{S}$ uma aproximação para a área da região $\mathcal{R}$ usando uma Soma de Riemann com 4 intervalos e escolhendo os pontos da direita.

(a) Calcule $\mathcal{S}$.

Solução:

Devemos dividir o intervalo $[-1, 1]$ (de comprimento 2) em 4 subintervalos de mesmo comprimento (logo, cada um deles terá comprimento 1/2). Usando pontos da direita, as alturas dos retângulos serão $f\left(-\frac{1}{2}\right)$, $f(0)$, $f\left(\frac{1}{2}\right)$ e $f(1)$.

```
> restart
> f:=x->(x^3+x^2-6*x)/(x^2+1)+5
```

$$f := x \mapsto \frac{x^3 + x^2 - 6 \cdot x}{x^2 + 1} + 5 \tag{16}$$

```
> S:=1/2*(f(-1/2)+f(0)+f(1/2)+f(1))
```

$$S := \frac{46}{5} \tag{17}$$

```
> evalf(S)
```

$$9.200000000 \tag{18}$$

(b) Verdadeiro ou falso? $\mathcal{S}$ é uma aproximação para a área da região $\mathcal{R}$ com erro menor que 10^{-1}. Justifique sua resposta.

Solução:

O valor exato da área de R é dado por $\int_{-1}^{1} f(x)\, \mathrm{d}x$.

```
> Area:=int(f(x),x=-1..1)
```

$$Area := 12 - \frac{\pi}{2} \tag{19}$$

S é então uma aproximação para a área de R com erro dado por:

```
> erro:=abs(S-Area)
```

$$erro := \frac{14}{5} - \frac{\pi}{2} \tag{20}$$

```
> evalf(erro)
```

$$1.229203673 \tag{21}$$

o qual é maior do que $10^{-1} = 0.1$. Logo, a afirmação é falsa.