**MAT4161 - Cálculo a uma Variável -- 2023.2**

**Gabarito P1 Maple - Versão I**

---

## Questão 1

Seja $k \in \mathbb{R}$ uma constante. Considere a função quadrática

$$f(x) = 2k\,x^2 + (2-9k)x + 8k\,.$$

(a) Encontre dois valores de $k$ para os quais o gráfico da função $f$ intersecta o eixo $x$ uma única vez.

*Obs.: Neste item, os valores devem ser exatos, não aproximados.*

```
> f:=x->2*k*x^2+(2-9*k)*x+8*k
```

$$f := x \mapsto 2\,k\,x^2 + (2-9\,k)\,x + 8\,k \qquad \textbf{(1)}$$

```
> a:=2*k; b:=2-9*k; c:=8*k
```

$$a := 2\,k$$
$$b := 2 - 9\,k$$
$$c := 8\,k \qquad \textbf{(2)}$$

```
> b^2-4*a*c
```

$$-64\,k^2 + (2-9\,k)^2 \qquad \textbf{(3)}$$

```
> solve(b^2-4*a*c=0)
```

$$2,\ \frac{2}{17} \qquad \textbf{(4)}$$

Logo, os valores pedidos são $k\ = 2$, $k\ = \dfrac{2}{17}$.

(b) Encontre um valor de $k$ para o qual a reta tangente ao gráfico de $f$ em $x_0 = 3$ é paralela à reta de equação $y = 6x - 4$.

*Obs.: Neste item, o valor deve ser exato, não aproximado.*

```
> D(f)(3)
```

$$3\,k + 2 \qquad \textbf{(5)}$$

```
> solve(D(f)(3)=6)
```

$$\frac{4}{3} \qquad \textbf{(6)}$$

Logo, o valor pedido é $k = \dfrac{4}{3}$.

(c) Encontre um valor de $k$ para o qual $y = -5x + 4$ é a equação da reta tangente ao gráfico de $f$ em algum $x_0$.

*Obs.: Neste item, o valor pode ser exato ou aproximado. Caso opte pelo valor aproximado, este deve ser dado com três casas decimais corretas.*

A equação da reta tangente ao gráfico de $f$ em $x = x_0$ é
$y = f'(x_0) \cdot (x - x_0) + f(x_0) = f'(x_0) \cdot x - f'(x_0) \cdot x_0 + f(x_0)$.

Para que esta equação seja igual a $y = -5\,x + 4$, devem valer as seguintes igualdades:

$f'(x_0) = -5$
$-f'(x_0) \cdot x_0 + f(x_0) = 4$

Substituindo a primeira equação na segunda, reescrevemos o sistema da seguinte forma:

$f'(x_0) = -5$
$-(-5) \cdot x_0 + f(x_0) = 4 \Rightarrow f(x_0) = -5 \cdot x_0 + 4$

Resolvendo o sistema com o Maple:

```
> solve([D(f)(x0)=-5,f(x0)=-5*x0+4])
```

$$\left\{k = RootOf\left(17\,\_Z^2 - 94\,\_Z + 49\right),\, x0 = \frac{17\, RootOf\left(17\,\_Z^2 - 94\,\_Z + 49\right)}{28} - \frac{31}{28}\right\} \qquad (7)$$

Para obter um valor aproximado para $k$, podemos usar o comando *fsolve*:

```
> fsolve([D(f)(x0)=-5,f(x0)=-5*x0+4])
```

$$\{k = 0.5826780011,\, x0 = -0.7533740708\} \qquad (8)$$

E então a solução seria --- com 3 casas decimais corretas --- $k = 0.582$.

Para obter soluções exatas, há dois caminhos a seguir.

Primeiro modo:

```
> solve(17*x^2-94*x+49=0)
```

$$\frac{47}{17} + \frac{4\sqrt{86}}{17},\ \frac{47}{17} - \frac{4\sqrt{86}}{17} \qquad (9)$$

Estes seriam os dois valores exatos de $k$ que satisfazem as condições do enunciado.

Segundo modo (usando comando em geral não visto em sala de aula):

```
> allvalues(solve([D(f)(x0)=-5,f(x0)=-5*x0+4]))
```

$$\left\{k = \frac{47}{17} - \frac{4\sqrt{86}}{17},\, x0 = \frac{4}{7} - \frac{\sqrt{86}}{7}\right\},\ \left\{k = \frac{47}{17} + \frac{4\sqrt{86}}{17},\, x0 = \frac{4}{7} + \frac{\sqrt{86}}{7}\right\} \qquad (10)$$

**Questão 2**

Considere $f : [-3, 2] \to \mathbb{R}$ a função dada por

$$f(x) = -2x^3 + 7x - 11\,.$$

(a) Determine as coordenadas do ponto do gráfico de $f$ que está mais próximo do ponto $P = (-1, 1)$. Justifique sua resposta.

*Obs.: Atenção ao domínio da função!*

```
> restart
> f:=x->-2*x^3+7*x-11
```

$$f := x \mapsto -2\,x^3 + 7\,x - 11 \qquad \textbf{(11)}$$

```
> d:=x->sqrt((x-(-1))^2+(f(x)-1)^2)
```

$$d := x \mapsto \sqrt{(x+1)^2 + (f(x)-1)^2} \qquad \textbf{(12)}$$

```
> plot(d(x),x=-3..2)
```




```
> solve(D(d)(x)=0)
```

$RootOf(12\_Z^5 - 56\_Z^3 + 72\_Z^2 + 50\_Z - 83, index = 1), RootOf(12\_Z^5 - 56\_Z^3 + 72\_Z^2 + 50\_Z - 83, index = 2), RootOf(12\_Z^5 - 56\_Z^3 + 72\_Z^2 + 50\_Z - 83, index = 3),$ **(13)**

$RootOf(12\_Z^5 - 56\_Z^3 + 72\_Z^2 + 50\_Z - 83, index = 4), RootOf(12\_Z^5 - 56\_Z^3 + 72\_Z^2 + 50\_Z - 83, index = 5)$

Como estamos procurando uma solução específica (pelo plot acima, o valor de $x$ que minimiza a distância está no intervalo $[-3,-2]$) e o solve acima não forneceu facilmente as soluções exatas, vamos usar o comando fsolve:

```
> xmin:=fsolve(D(d)(x)=0,x=-3..-2)
```

$$xmin := -2.439173864 \tag{14}$$

```
> f(xmin)
```

$$0.94985005 \tag{15}$$

Logo, o ponto do gráfico de $f$ que está mais próximo do ponto $P = (-1, 1)$ possui coordenadas $(-2.439, 0.949)$.

(b) Determine a coordenadas do ponto do gráfico de $f$ que está <u>mais distante</u> do ponto $P = (-1, 1)$. Justifique sua resposta.

*Obs.: Atenção ao domínio da função!*

Pelo plot acima, vemos que o maior valor para a distância ocorre quando $x = -3$.

```
> xmax:=-3
```

$$xmax := -3 \tag{16}$$

```
> f(xmax)
```

$$22 \tag{17}$$

Logo, o ponto do gráfico de $f$ que está mais distante do ponto $P = (-1, 1)$ possui coordenadas $(-3, 22)$.